

令和3年度 学士論文

副作用を持つ
プログラミング言語を
型安全に定義するための
ライブラリの開発に向けて

東京工業大学 情報理工学院数理・計算科学系

学籍番号 18B10155

津山勝輝

指導教員

増原 英彦 教授

令和3年2月1日

概要

型安全インタプリタ [1] とはプログラミング言語の意味論仕様の実装方法のひとつである。型安全インタプリタの重要な特徴は、メタ言語の型検査を通して対象言語の型安全性を証明できることである。対象言語を実用的な言語機能で拡張するとき、型安全インタプリタに証明項を追加して、その機能が型安全に動作することを証明する必要がある。こうした証明項はインタプリタの実装を繁雑にするという問題がある。本研究の目的は、エフェクトやコエフェクトを持つ言語の型安全インタプリタの簡潔な実装を支援するライブラリの開発である。本論文では、特にエフェクト言語の型安全インタプリタの実装に有効な抽象化を分析するため、限定継続命令 (shift/reset) を持つ言語 $\lambda_{s/r}$ と、代数的エフェクトとハンドラを持つ言語 λ_e [16] の型安全インタプリタを依存型言語 Agda [11] を使って実装した。抽象構文木の変数束縛に PHOAS [6] を用い、インタプリタを継続渡しスタイル (CPS) で実装することで、CPS 意味論の書き下しとして型安全インタプリタを実装できる。

謝辞

本研究を進めるにあたり、増原英彦教授、叢悠悠助教授に多くのアドバイスやご指導をいただきました。また、増原研究室の学生の皆様にも様々な知見や、研究におけるアドバイスをいただきました。本論文は以上の方々のご支援がなければ存在しえませんでした。この場を借りて感謝申し上げます。

Contents

第 1 章	はじめに	1
第 2 章	実装例：単純型付き入計算	3
2.1	単純型付き入計算	3
2.2	CPS 意味論	3
2.3	構文の実装	4
2.4	型の変換の実装	7
2.5	インタプリタの実装	7
第 3 章	限定継続命令による拡張	10
3.1	継続	10
3.2	答えの型	11
3.3	構文の定義	11
3.4	CPS 意味論とインタプリタ	13
第 4 章	代数的エフェクトによる拡張	16
4.1	代数的エフェクトとハンドラ	16
4.2	構文	17
4.3	型規則	18
4.4	CPS 意味論	19
4.5	構文の実装	20
4.6	型の変換	22
4.7	インタプリタ	23
第 5 章	関連研究	25
5.1	型付き構文木と型安全インタプリタ	25
5.2	高度な機能の型安全な実装	25
第 6 章	まとめと今後の課題	27
6.1	トレースエフェクト	28
6.2	コエフェクト計算	28

List of Figures

2.1	STLC の型規則	4
2.2	STLC の CPS 意味論	5
2.3	型の構文実装	5
2.4	型の変換の定義	7
3.1	$\lambda_{s/r}$ の構文	12
3.2	$\lambda_{s/r}$ の型規則	13
3.3	$\lambda_{s/r}$ の意味論	14
4.1	λ_e の構文	18
4.2	λ_e の型規則	19
4.3	λ_e の CPS 意味論	20

List of Tables

第1章 はじめに

あるプログラミング言語のインタプリタを別のプログラミング言語で実装するとき、対象言語 (実装される側の言語) の意味論の仕様をメタ言語 (実装する側の言語) の上で定義することになる。対象言語の意味論を、メタ言語で定義することを目的に書かれるインタプリタを、**定義インタプリタ**という。

対象言語が型付きである場合、その型システムを実装する方法は複数ある。一つは、インタプリタとは別に型検査器を実装するという方法がある。このとき、型安全性、すなわち型検査器を通過した対象言語のプログラムに対して、インタプリタは未定義の動作をしないという性質が重要になる。型検査器を別個に作成する方法では、型安全性の証明を得るために、Coq や Agda などによる形式化が必要であり、これが言語実装者にとっての負担であった。

これを解決するために提唱されたのが、**型安全インタプリタ** (intrinsically-typed interpreter)[1] である。型安全インタプリタは、定義インタプリタの実装方法の一種である。型安全インタプリタの利点は主に以下の二つである。

- 実行可能な形式なので、意味論のテストが可能である。
- インタプリタがメタ言語の型検査を通過すれば、対象言語の型安全性が証明される。

特に重要なのは2つ目である。メタ言語上の型安全インタプリタの型で対象言語の型安全性の言明を表すことで、カーリー=ハワード同型により、型安全インタプリタの本体は、対象言語の型安全性の証明とみなすことができる。

対象言語を実用的な言語機能で拡張するとき、型安全インタプリタに証明項を追加して、その機能が型安全に動作することを証明する必要がある。こうした証明項はインタプリタの実装を複雑にするという問題がある。先行研究 [3, 14] は、インタプリタの実装を抽象化し、証明項を隠すことでこの問題に対処している。例えば、Poulsen ら [3] は、可変状態を持つ言語の型安全インタプリタを実装するためのライブラリを開発した。この言語の型安全インタプリタは、可変状態の数が評価に従って増えるこ

とを示す証明項を持つ。しかしこうした証明項は意味論実装に直接関係がなく、これらを明示的に扱えばインタプリタ定義は繁雑になってしまう。この問題を解決するために、Poulsen らはモナドを利用した抽象化で証明項をカプセル化し、インタプリタの簡潔な実装をサポートしている。

本研究の目的は、エフェクトやコエフェクトを持つ言語のインタプリタを簡潔に実装するためのライブラリを開発することである。本論文は特にエフェクトに注目し、エフェクト言語の型安全インタプリタにどのような抽象化が有効かを分析するため、限定継続命令 (shift/reset) を持つ言語 $\lambda_{s/r}$ と、代数的エフェクトとハンドラを持つ言語 λ_e [16] の型安全インタプリタを実装した。実装には依存型言語の Agda [11] を用い、2つの言語の型付き抽象構文木の変数束縛には PHOAS [6] を用いた。また、 $\lambda_{s/r}$ 、 λ_e は明示的に継続を操作する機構があるため、インタプリタは CPS で記述した。

- $\lambda_{s/r}$ の型付き抽象構文木は、実行前と実行後の文脈が返す型の2つを保持する。型安全インタプリタは、shift/reset のための CPS 意味論 [8] に基づいて実装した。
- λ_e の型付き抽象構文木は、項が実行する可能性のあるオペレーションの集合を保持する。代数的エフェクトとハンドラのための CPS 変換 [9] に基づき、 λ_e の CPS 意味論を構築し、それに対応する CPS インタプリタを記述した。

また、抽象構文木定義に PHOAS を使うことによる利点は主に以下の2つである。

- 対象言語における変数への代入がメタ言語の関数適用によって型安全に行われるので、メタ言語で代入規則を定義する必要がない。
- インタプリタの継続と、対象言語の関数が同じメタ言語レベルで扱われるため、継続の操作のために特別な命令をメタ言語で定義する必要がなく、CPS 意味論さえあれば、それを直接書き下す形でインタプリタを記述できる。

本論文の構成は以下の通りである。2章で、STLC のための型安全インタプリタの具体的な実装を通じて、本研究のアプローチについて説明する。3章で、STLC を限定継続命令で拡張した言語のための実装を説明する。4章で、STLC を代数的エフェクトとハンドラで拡張した言語のための実装を説明する。??章で関連研究を紹介し、最後に ??章でまとめと今後の課題を述べる。

第2章 実装例：単純型付きλ計算

単純型付きλ計算は純粋関数型言語の小さなモデルである。この章では、単純型付きλ計算のための型安全インタプリタの実装を通じて、このアプローチについて説明する。

2.1 単純型付きλ計算

単純型付きλ計算の構文を以下に示す。

$$\begin{aligned}
 \text{(型)} \quad \tau &::= \text{Unit} \mid \tau \rightarrow \tau \\
 \text{(項)} \quad t &::= \text{unit} \mid x \mid \lambda x.t \mid t t \\
 \text{(値)} \quad v &::= \text{unit} \mid \lambda x.t \\
 \text{(型環境)} \quad \Gamma &::= \emptyset \mid \Gamma, x : t
 \end{aligned}$$

型は *Unit* 型と関数型からなる。項は変数 x 、ラムダ抽象 $\lambda x.t$ 、関数適用 $t t$ と、*unit* からなる。値はラムダ抽象 $\lambda x.t$ と *unit* のみからなり、項はこれらの値のうちのどれかに評価されることになる。

型環境 Γ は、束縛された変数と、それが参照する値の型のペアの集合である。型付け判断 $\Gamma \vdash t : \tau$ は型環境 Γ のもとで項 t は τ 型を持つという意味である。型規則は、型付け判断を導出するための推論規則である。STLC の型規則を図 2.1 に示す。規則 T-UNIT は *unit* がどの型環境においても *Unit* 型を持つことを表す。規則 T-VAR は、変数が型環境 Γ で束縛されているならば、対応する型をその変数の型にする規則である。規則 T-ABS は、関数 $\lambda x.t$ に型を付ける規則である。関数本体 t が、引数の型 $x : \tau_1$ を加えた型環境の元で τ_2 型になるとき、 $\lambda x.t$ は $\tau_1 \rightarrow \tau_2$ 型を持つ。規則 T-APP は $\tau_1 \rightarrow \tau_2$ 型の関数に τ_1 の項を適用すれば τ_2 型になることを表す。

2.2 CPS 意味論

この節では、STLC の意味論を継続渡しスタイル [12] で記述する。意味論は、STLC の項がどのように評価されるかを表す。ここでは、評価戦略として左から右への値呼びをする。CPS で記述された評価関数は、「現在

型規則

$$\begin{array}{c}
\Gamma \vdash \mathit{unit} : \mathit{Unit} \qquad\qquad\qquad (\text{T-UNIT}) \\
\\
\frac{x : \tau \in \Gamma}{\Gamma \vdash x : \tau} \qquad\qquad\qquad (\text{T-VAR}) \\
\\
\frac{\Gamma, x : \tau_1 \vdash t : \tau_2}{\Gamma \vdash \lambda x. t : \tau_1 \rightarrow \tau_2} \qquad\qquad\qquad (\text{T-ABS}) \\
\\
\frac{\Gamma \vdash t_1 : \tau_1 \rightarrow \tau_2 \quad \Gamma \vdash t_2 : \tau_1}{\Gamma \vdash t_1 t_2 : \tau_2} \qquad\qquad\qquad (\text{T-APP})
\end{array}$$

Figure 2.1: STLC の型規則

行っている評価の結果の値を受け取り、残りの評価を行う関数」を引数に受け取ることによって評価順序を規定する。この残りの評価を行う関数のことを継続というが、継続については3章で詳細に説明する。

STLC の CPS 意味論を図 2.2 に示す。 $\xi[e]\rho\kappa$ は「環境 ρ と継続 κ のもとで項 e を評価した結果」を表す。環境 ρ は、変数が参照する値の集合であり、継続 κ は、残りの評価を表す。 unit の評価は、継続 κ に unit を渡し計算を再開させる。変数 x の評価は、環境 ρ から対応する値を探し出し、その値を継続に与える。関数 $\lambda x. t$ は、メタ言語における関数へ評価される。その関数は引数の値 v と新しい継続 κ' を受け取り、 x の値を現在の環境に追加して κ' のもとで関数本体 t を評価する。関数適用 $t_1 t_2$ の評価は、 t_1 、 t_2 の順に評価し、最後にそれぞれの結果 f 、 v について、 f に v と現在の継続 κ を渡す。型規則より、 f はラムダ抽象の評価結果であるはずなので、 $f v \kappa$ がどのように処理されるかを理解するには、 $\xi[\lambda x. e]\rho\kappa$ のケースを見るとよい。これによると、 $f v \kappa$ は、環境に v を追加し、関数 t_1 の本体を κ のもとで評価する。関数適用の評価順序は ξ に渡す継続によって制御している。よってメタ言語の評価順序に依存しない。

次の節からは、STLC のための型付き抽象構文木と型安全インタプリタの実装を示す。

2.3 構文の実装

まず、型の定義にしたがって STLC の型を表す Agda のデータ型 `Ty` を図 2.3 に定義する。各コンストラクタは、 unit 型と関数型を表す。関数型

$$\begin{aligned}
\xi[unit]\rho\kappa &= \kappa(unit) \\
\xi[x]\rho\kappa &= \kappa(\rho(x)) \\
\xi[\lambda x.e]\rho\kappa &= \kappa(\lambda v.\lambda\kappa'.\xi[e]\rho[v/x]\kappa') \\
\xi[e1\ e2]\rho\kappa &= \xi[e1]\rho(\lambda f.\xi[e2]\rho(\lambda v.f\ v\ \kappa))
\end{aligned}$$

Figure 2.2: STLC の CPS 意味論

```

data Ty : Set where
  Unit : Ty
  _⇒_ : Ty → Ty → Ty

```

Figure 2.3: 型の構文実装

のコンストラクタは「引数型と返り値型の2つを受け取り、型になる」と読むことができる。

次に、型付き抽象構文木を以下に定義する。

```

data Expr (Var : Ty → Set) : Ty → Set where

```

型付き抽象構文木の Agda における型は、STLC の型判断の言明 $\Gamma \vdash t : T$ を表す。すなわち、`Expr Var T` は、STLC の T 型の項を表す型である。`Expr` は `Ty` 型の値を受け取って型になる。これは Agda が持つ依存型の特徴である。

AST の設計にあたり、変数の名前解決には PHOAS[6] を使用する。`Expr` 型が受け取る `Var` は PHOAS 特有のパラメータで、変数が参照する値の型を表す。これによって、無限再帰¹ を起こすことなく、メタ言語の束縛

¹PHOAS の前身の HOAS では、以下のような定義になる。

```

data Expr : Set where
  lam : (Expr → Expr) → Expr
このような定義は Agda に許容されない。このコンストラクタによって以下のような無限再帰を起こすプログラムが書けるからである。
p : Expr → Expr → Expr
p (lam f) = f
w : Expr → Expr
w x = p x x

```

機構を使って、オブジェクト言語の変数を管理できる。

`Expr` 型の各コンストラクタ定義を、STLC の型規則にしたがって実装する。ここからは、`Expr` 型のコンストラクタを、対応する型規則と共に示す。特に、これらのコンストラクタの型は、引数型が型規則の仮定を表し、返り値型が結論を表すということに注目すべきである。

`unit` コンストラクタは、型規則 T-UNIT を表し、自明に `Unit` 型を持つ。

$$\text{unit} : \text{Expr} \text{ Var Unit} \quad \Gamma \vdash \text{unit} : \text{Unit} \text{ (T-UNIT)}$$

`var` コンストラクタは、参照先の `Var T` 型の値を受け取って `T` 型の変数を表す。`Var T` 型の値は、STLC における `T` 型の値に対応する。このコンストラクタの型は型規則 T-VAR を表す。この規則の仮定 $x : \tau \in \Gamma$ は、`var` コンストラクタが参照先の値を引数に受け取っていることで保証される。 $\forall \{T\}$ で囲まれた変数 `T` は明示的に与える必要はなく、Agda によって推論させる。

$$\text{var} : \forall \{T\} \rightarrow \text{Var } T \rightarrow \text{Expr} \text{ Var } T$$

$$\frac{x : \tau \in \Gamma}{\Gamma \vdash x : \tau} \quad \text{(T-VAR)}$$

`lam` コンストラクタは、「`A` 型の値を受け取り `B` 型の項を返す関数」を受け取って、`A \Rightarrow B` 型の項になる。このコンストラクタが受け取る関数は、メタ言語 (ここでは Agda のこと) の関数であることに注意すべきである。対象言語 (ここでは STLC) の変数束縛を、メタ言語の変数束縛で表現している。これは PHOAS の特徴である。

$$\text{lam} : \forall \{A B\} \rightarrow (\text{Var } A \rightarrow \text{Expr} \text{ Var } B) \rightarrow \text{Expr} \text{ Var } (A \Rightarrow B)$$

$$\frac{\Gamma, x : \tau_1 \vdash t : \tau_2}{\Gamma \vdash \lambda x. t : \tau_1 \rightarrow \tau_2} \quad \text{(T-ABS)}$$

`app` コンストラクタは関数適用を表す。このコンストラクタは、`A \Rightarrow B` 型の項と `A` 型の項を受け取り `B` 型の項になる。

$$\text{app} : \forall \{A B\} \rightarrow \text{Expr} \text{ Var } (A \Rightarrow B) \rightarrow \text{Expr} \text{ Var } A \rightarrow \text{Expr} \text{ Var } B$$

$$\frac{\Gamma \vdash t_1 : \tau_1 \rightarrow \tau_2 \quad \Gamma \vdash t_2 : \tau_1}{\Gamma \vdash t_1 t_2 : \tau_2} \quad \text{(T-APP)}$$

$$\begin{aligned} \sim t &: \text{Ty} \rightarrow \text{Set} \\ \sim t \text{ Unit} &= \top \\ \sim t (A \Rightarrow B) &= \forall \{ \alpha : \text{Set} \} \rightarrow (\sim t A) \rightarrow (\sim t B \rightarrow \alpha) \rightarrow \alpha \end{aligned}$$

Figure 2.4: 型の変換の定義

このように、型付き抽象構文木は型システムの仕様を表現する。

2.4 型の変換の実装

この章の最終目的は、型付き抽象構文木に対する型安全インタプリタを定義することである。インタプリタは STLC の項を Agda の値へ評価するように設計する。したがって、インタプリタ実装のために STLC の型を Agda の型に対応させる関数が必要である。型の変換 $\sim t$ の実装を図 2.4 に示す。 \top は Agda 上の Unit 型と解釈してよい。STLC の関数型の変換結果は、 $\sim t A$ 型の引数と $\sim t B \rightarrow \alpha$ 型の関数で表される、関数適用後の継続を受け取り、 α 型の値を返す Agda の関数型である。 $\sim t$ の本来の型は $\text{Ty} \rightarrow \text{Set}_1$ であるが、Agda の `{-# -type-in-type #-}` オプションを使うことで Set と Set_1 の違いを無視している。

2.5 インタプリタの実装

これまでの定義を使うことで、型安全インタプリタを記述できる。まず、インタプリタの型定義を以下に示す。

$$\begin{aligned} \text{eval} &: \forall \{ T \} \rightarrow \text{Expr } \sim t T \rightarrow \sim t T \\ \text{evale} &: \forall \{ T \alpha \} \rightarrow \text{Expr } \sim t T \rightarrow (\sim t T \rightarrow \alpha) \rightarrow \alpha \end{aligned}$$

型安全インタプリタ `eval` は、トップレベルの項に対する評価関数であり、 T 型の項を受け取り、 $\sim t T$ 型の値へ評価する。これは、STLC の型安全性「 T 型の項は、 T 型の値へ評価される」を表している。したがって、`eval` 関数がホスト言語の型検査を通れば、STLC の型安全性を証明したことになる。`eval` の本体は、`evale` を用いて実装される。`evale` は T 型の項と、 $\sim t T \rightarrow \alpha$ 型の継続を受け取り、 α 型の値へ評価する関数である。 α は、継続が最終的に返す値の型を表す。`eval` の本体は以下である。`eval` は与えられた項を空の継続 (`id` 関数と等価) のもとで評価する。

$$\text{eval } e = \text{evale } e (\lambda x \rightarrow x)$$

`evale`の実装はCPS意味論に基づいている。CPS意味論の評価関数 ξ を`evale`とし、 ξ が行う評価計算をAgdaで書き下すことで、`evale`の実装が得られる。ここからは、各コンストラクタに対する`evale`の動作を対応するCPS意味論と共に示す。特に、CPS意味論の式とインタプリタのプログラムが類似していることに注目すべきである。

`unit`コンストラクタは、Agdaの \top 型の値である`tt`に評価される。なので、`evale`は継続 k を`tt`で再開させる。 ξ が受け取る継続 κ が、Agdaにおける`evale`の引数 k として意味論と同様に扱える点に注目すべきである。

$$\text{evale } \text{unit } k = k \text{ tt}$$

$$\xi[\text{unit}]\rho\kappa = \kappa(\text{unit})$$

`var`コンストラクタの評価は、単に参照先の値を返す。CPS意味論の、環境から変数の参照先の値を取る $\rho(x)$ は、`var`コンストラクタから引数 x を取り出す操作に相当する。よって`evale`は環境 ρ に対応する引数を必要としない。

$$\text{evale } (\text{var } x) k = k x$$

$$\xi[x]\rho\kappa = \kappa(\rho(x))$$

`lam`コンストラクタの評価結果は「引数の値を受け取り、ラムダ抽象の本体に代入して生成される項を評価する関数」である。CPS意味論の、環境に変数を追加する操作 $\rho[v/x]$ は、`lam`コンストラクタの引数で、関数本体を表す f を v に適用することに相当する。PHOASによる定義によって、変数への値の代入をAgdaの関数適用で表現でき、別途の代入規則などを定義する必要がなくなっている。

$$\text{evale } (\text{lam } f) k = k \$ \lambda v k' \rightarrow \text{evale } (f v) k'$$

$$\xi[\lambda x.e]\rho\kappa = \kappa(\lambda v.\lambda\kappa'.\xi[e]\rho[v/x]\kappa')$$

`app`コンストラクタの評価は、これまでのコンストラクタと同様に、CPS意味論の書き下しとして見ることができる。通常の型システムを持つ言語では、 f が関数でないことによるインタプリタの例外を考慮する必要がある。依存型言語のAgdaは、依存型パターンマッチングによって、 e_1 の評価後の型が $t A \rightarrow t B$ であることを保証するため、このような例外を除外できる。これは、メタ言語に依存型言語を使うことによる利点のひとつである。

```

evale (app e1 e2) k =
  evale e1 $ λ f →
  evale e2 $ λ v →
  f v k

```

$$\xi[e1\ e2]\rho\kappa = \xi[e1]\rho(\lambda f.\xi[e2]\rho(\lambda v.f\ v\ \kappa))$$

以上のように、型付き構文木を PHOAS で定義することによって、CPS 意味論の書き下しとして、インタプリタを定義することができる。以降の章では、限定継続命令や代数的エフェクトとハンドラを持つ言語についての、この方法による実装を説明する。

第3章 限定継続命令による拡張

この章では、STLC を限定継続命令 `shift/reset`[8] で拡張した言語を実装する方法を説明する。まず、3.1 節、3.2 節で、継続の概念と答えの型について説明する。次に 3.3 節で対象言語の構文と型システムを定義し、Agda による構文実装を示す。最後に 3.4 節で CPS 意味論とインタプリタ実装を示す。

3.1 継続

継続とは「ある時点での残りの計算」を表す概念である。具体的には、ある計算 $5 * (2 + 3) - 4$ において、部分式 $(2 + 3)$ の継続は $5 * x - 4$ である。ただし、 x は $(2 + 3)$ の計算結果で置き換えられる。また、限定継続は、継続の範囲が定まっているものである。例えば、継続の範囲を $\langle \cdot \rangle$ で区切るとすると、計算 $5 * \langle (2 + 3) - 4 \rangle$ において、 $(2 + 3)$ の継続は $x - 4$ となる。

`shift/reset`[8] は、限定継続をプログラム中で明示的に扱うための命令である。`reset` は、継続の範囲を区切る命令である。`reset(e)` で囲まれた式 e の意味は変わらないが、継続は `reset` の内部までに限定される。例えば、 $5 * \text{reset}((2 + 3) - 4)$ において、 $(2 + 3)$ の継続は $x - 4$ である。

ここまでに例示した継続は「引数 x を受け取り、残りの計算を評価する関数」と読むこともできる。もうひとつの命令 `shift` は、その時点での継続を関数として取る命令である。`shift($\lambda k.e$)` は評価される時点の継続を関数として変数 k に束縛して式 e を評価する。例えば、 $5 * \text{shift}(\lambda k.k(2 + 3)) - 4$ について、変数 k に束縛される継続は $5 * x - 4$ である。この計算の評価過程を以下に示す。

```
reset(5 * shift(fun k -> k (2 + 3)) - 4)
--> reset((fun k -> k (2 + 3)) (fun x -> reset(5 * x - 4)))
--> reset((fun x -> reset(5 * x - 4)) (2 + 3))
--> reset(5 * (2 + 3) - 4)
```

`shift/reset` には様々な応用がある。ここでは、非決定計算を `shift/reset` を使って記述する例を示す。


```
coin () = shift (fun k -> [k true , k false])
```

関数 *coin* は、自身の呼び出しを *true* と *false* の2通りに置き換えた上で、残りの計算を評価し、それぞれの結果をリストにする関数である。関数としての継続を異なる値にたいして2回呼び出すことで非決定性が表現されている。具体的には以下のように動作する。

```
reset( if coin() then 1 else 0 )
--> (fun k -> [k true , k false]) (fun x -> if x then 1 else 0)
--> [if true then 1 else 0 , if false then 1 else 0]
--> [1 , 0]
```

3.2 答えの型

STLC を *shift/reset* で拡張すると、型システムに新たに答えの型という概念が必要になる。答えの型とは「現時点の継続が返す型」である。例えば、 $(2 + 2) == 5$ において、 $(2 + 2)$ を評価するときの継続は $x == 5$ であり、その継続の返り値の型は *Bool* 型であるので、 $(2 + 2)$ の答えの型は *Bool* 型である。このように評価の前後で答えの型が変化しない計算を、「純粋である」と言う。STLC の項に限れば、評価の前後で答えの型が変わることはないが、*shift/reset* で拡張したとき、答えの型が変化する項を書くことができる。例えば、先ほど示した関数 *coin* について、以下の式で *coin()* を評価する直前の継続は *Int* 型を返すが、*coin()* を評価したあとは、*List Int* 型を返すようになる。これが答えの型の変化である。型システムはこの答えの型の変化を検出する必要がある。

```
coin () = shift (fun k -> [k true , k false])
reset( if coin() then 1 else 0 )
```

3.3 構文の定義

この章で扱う、STLC を *shift/reset* で拡張した言語を、本稿では $\lambda_{s/r}$ と呼ぶことにする。 $\lambda_{s/r}$ の構文定義を図 3.1 に示す。 $t_1/\alpha \rightarrow t_2/\beta$ は t_1 から t_2 への関数で、評価すると答えの型が α から β へ変化することを示す。項は STLC の項に *shift* と *reset* を追加したものである。

ここで、型の定義にしたがって、型の構文を Agda で以下のように定義できる。

```
data Ty : Set where
  Unit : Ty
  ./->./- : Ty -> Ty -> Ty -> Ty -> Ty
```

(型) $\tau ::= \text{Unit} \mid \tau/\tau \rightarrow \tau/\tau$
 (項) $t ::= \text{unit} \mid x \mid \lambda x.t \mid t t \mid \text{shift}(\lambda k.t) \mid \text{reset}(t)$

Figure 3.1: $\lambda_{s/r}$ の構文

型判断は $\Gamma; \alpha \vdash e : \tau; \beta$ の形をとる。これは、「型環境 Γ のもとで項 e は τ 型を持ち、 e を評価すると答えの型が α から β へ変化する」ことを意味する。この型判断に対応するため、Agda の実装において、型付け可能な項の型である `Expr` 型は評価前と評価後の答えの型に依存するよう以下のように設計する。

```
data Expr (Var : Ty → Set) (α : Ty) : Ty → Ty → Set where
  -- ...
```

`Expr Var α T β` 型は、 $\lambda_{s/r}$ において $\Gamma; \alpha \vdash e : T; \beta$ と型付けされる項を表す型である。

次に、 $\lambda_{s/r}$ 型規則を図 3.2 に示す。型規則の定義は、Danvy と Filinski ら [7] の型システムに基づいている。純粋な項の型規則では、評価の前後で answer type が同じであることに注目すべきである。規則 T-RESET は、`reset` で囲まれた式 t の評価後の答えの型 τ が、`reset(t)` の型になることを表す。規則 T-SHIFT は、文脈内で `shift(λk.t)` に期待される型 τ と文脈の答えの型 α で継続に型を付けて k に束縛し、その上で t の型検査を行う。

STLC と同様、型規則にもとづいて、`Expr` 型のコンストラクタを以下のように定義することができる。

```
unit : Expr Var α Unit α
var : ∀ {T} → Var T → Expr Var α T α
fun : ∀ {A B β γ} →
      (Var A → Expr Var β B γ) →
      Expr Var α (A / β ⇒ B / γ) α
app : ∀ {A B β γ δ} →
      Expr Var γ (A / α ⇒ B / β) δ →
      Expr Var β A γ →
      Expr Var α B δ

reset : ∀ {T β} → Expr Var β β T → Expr Var α T α

shift : ∀ {T β γ δ} →
```

型規則

$$\begin{array}{c}
\frac{x : \tau \in \Gamma}{\Gamma; \alpha \vdash x : \tau; \alpha} \quad (\text{T-VAR}) \\
\\
\frac{\Gamma, x : \tau_1; \alpha \vdash t : \tau_2; \beta}{\Gamma; \gamma \vdash \lambda x. t : \tau_1 / \alpha \rightarrow \tau_2 / \beta; \gamma} \quad (\text{T-ABS}) \\
\\
\frac{\Gamma; \gamma \vdash t_1 : \tau_1 / \alpha \rightarrow \tau_2 / \beta; \delta \quad \Gamma; \beta \vdash t_2 : \tau_1; \gamma}{\Gamma; \alpha \vdash t_1 t_2 : \tau_2; \gamma} \quad (\text{T-APP}) \\
\\
\frac{\Gamma, k : \tau / \delta \rightarrow \alpha / \delta; \gamma \vdash t : \tau; \beta}{\Gamma; \alpha \vdash \mathit{shift}(\lambda k. t) : \tau; \beta} \quad (\text{T-SHIFT}) \\
\\
\frac{\Gamma; \alpha \vdash t : \alpha; \tau}{\Gamma; \beta \vdash \mathit{reset}(t) : \tau; \beta} \quad (\text{T-RESET})
\end{array}$$

Figure 3.2: $\lambda_{s/r}$ の型規則

$$\begin{array}{l}
(\text{Var } (T / \gamma \Rightarrow \alpha / \gamma) \rightarrow \text{Expr } \text{Var } \delta \delta \beta) \rightarrow \\
\text{Expr } \text{Var } \alpha T \beta
\end{array}$$

3.4 CPS 意味論とインタプリタ

図 3.3 に CPS 意味論を示す。純粹な項の評価は、最終的に、現在の継続に評価結果の値を渡すことで計算を再開させる。限定継続命令の評価では、評価時の継続の取り扱いが STLC と比べて変化する。 $\mathit{reset}(e)$ の評価は、 e を空の継続のもとで評価し、その結果を現在の継続に渡して計算を再開する。 $\mathit{shift}(\lambda k. e)$ の評価は、現在の継続をメタ言語上の関数として k に束縛し、 e に代入したものをさらに評価する。 e を評価するときの継続は空の継続 (なにもせず終了する計算) である。

型システムと意味論を定義したので、 $\lambda_{s/r}$ の型安全性を定義する。 $\lambda_{s/r}$ の型安全性は、 $\Gamma; \alpha \vdash e : \tau; \beta$ なる e に関して、 $\xi[e]_{\rho}(\lambda v. v)$ が β 型になることである。

ここで、CPS 意味論に基づく型安全インタプリタの実装を示す。まず、STLC の実装と同様に型の変換の定義が必要である。CPS 意味論で示した、評価の結果の型に対応させる。型の変換 \sim_t の実装を以下に示す。

$$\begin{aligned}
\xi[\mathit{unit}]_{\rho\kappa} &= \kappa(\mathit{unit}) \\
\xi[x]_{\rho\kappa} &= \kappa(\rho(x)) \\
\xi[\lambda x.e]_{\rho\kappa} &= \kappa(\lambda v.\lambda\kappa'.\xi[e]_{\rho[v/x]\kappa'}) \\
\xi[e_1 e_2]_{\rho\kappa} &= \xi[e_1]_{\rho}(\lambda f.\xi[e_2]_{\rho}(\lambda v.f v \kappa)) \\
\xi[\mathit{reset}e]_{\rho\kappa} &= \kappa(\xi[e]_{\rho} \mathit{id}) \\
\xi[\mathit{shift}(\lambda k.e)]_{\rho\kappa} &= \xi[e]_{\rho}[\lambda v.\lambda\kappa'.\kappa'(\kappa(v))/k]
\end{aligned}$$

Figure 3.3: $\lambda_{s/r}$ の意味論

$$\begin{aligned}
\sim t &: \mathbf{Ty} \rightarrow \mathbf{Set} \\
\sim t \mathbf{Unit} &= \top \\
\sim t (A / \alpha \Rightarrow B / \beta) &= \\
&(\sim t A) \rightarrow ((\sim t B) \rightarrow (\sim t \alpha)) \rightarrow (\sim t \beta)
\end{aligned}$$

次に、インタプリタの実装を示す。以下はインタプリタの型である。

$$\begin{aligned}
\mathit{eval} &: \forall \{ \alpha \beta \} \rightarrow \mathbf{Expr} \sim t \alpha \alpha \beta \rightarrow \sim t \beta \\
\mathit{evale} &: \forall \{ T \alpha \beta \} \rightarrow \mathbf{Expr} \sim t \alpha T \beta \rightarrow (\sim t T \rightarrow \sim t \alpha) \rightarrow \sim t \beta
\end{aligned}$$

STLC の実装と同様に、 eval はトップレベルの項のインタプリタであり、型安全インタプリタ evale は $\lambda_{s/r}$ の型安全性を表す。 eval が受け取る項の実行前の答えの型は項自身の型と一致する。 $(\sim t T \rightarrow \sim t \alpha)$ は継続の型である。 evale は T 型の式と現在の継続を受け取り、計算を行う。

インタプリタ本体の実装を以下に示す。 eval は項を空の継続のもとで評価する。STLC の実装と同様、 evale は CPS 意味論の仕様を反映している。

$$\begin{aligned}
\mathit{eval} e &= \mathit{evale} e (\lambda x _i x) \\
\mathit{evale} \mathit{unit} k &= k \mathbf{tt} \\
\mathit{evale} (\mathbf{var} x) k &= k \$ x \\
\mathit{evale} (\mathbf{fun} f) k &= k \$ \lambda x \rightarrow \lambda k' \rightarrow \mathit{evale} (f x) k' \\
\mathit{evale} (\mathbf{app} e_1 e_2) k &= \\
&\mathit{evale} e_1 \$ \lambda f \rightarrow \\
&\mathit{evale} e_2 \$ \lambda v \rightarrow \\
&f v k \\
\mathit{evale} (\mathbf{reset} e) k &=
\end{aligned}$$

$$k \text{ \$ } \text{evale } e (\lambda x \rightarrow x)$$
$$\text{evale } (\text{shift } f) k =$$
$$\text{evale } (f \text{ \$ } \lambda v k' \rightarrow k' \text{ \$ } k v) (\lambda x \rightarrow x)$$

対象言語に限定継続命令が追加された場合でも、適切な CPS 意味論と、PHOAS の型付き AST を与えれば、以上のように CPS 意味論に直接対応する型安全インタプリタを記述できる。この方法は `shift/reset` 以外の限定継続命令に対しても有効であると考えられる。

第4章 代数的エフェクトによる拡張

この章では、STLCを代数的エフェクト [13] によって拡張した言語を実装する方法について説明する。まず、4.1節で代数的エフェクトの概念について簡単に説明する。次に4.1節, 4.3節で構文と型規則を説明し、4.4節で意味論を説明する。最後に4.5～節4.7でAgdaによる実装を示す。

4.1 代数的エフェクトとハンドラ

代数的エフェクトとは、例外・状態などの副作用を表すための機能である [13]。代数的エフェクトを持つ言語では、オペレーション呼び出しによって副作用を発生させ、副作用はハンドラによって処理される。

代数的エフェクトを用いたプログラム例を示す。ここでは、3章で例示した非決定性計算を扱う。非決定性計算を表すエフェクト $NDet$ と、そのオペレーション $coin$ は Eff 言語 [4] に似た構文を使って以下のように定義できる。

```
effect NDet {
  control coin() : bool
}
```

オペレーション $coin$ の定義は、「呼び出されると $Bool$ 型の値を返す」と読むことができる。ただし、この時点で $coin$ の振る舞いは定義されていないことに注意すべきである。 $coin$ を用いた計算の例を以下に示す。

```
handle {
  if (do coin()) then (return 1) else (return 0)
}
with {
  return x → x
  coin () k → [k true , k false]
}
```

これは、 $handle$ の後続くオペレーション呼び出しを含む計算を、 $with$ の後続くハンドラのもとで実行するプログラムである。ここでのハンド

ラの定義によって、*coin* オペレーションが呼び出されたときには、その時点での継続を *true*、*false* の2通りの値で再開させた結果をリストにして返すようになる。この計算の実行過程を以下に示す。

```
if (do coin()) then (return 1) else (return 0)
--> (fun _ k -> [k true , k false])
      () (fun x -> if x then (return 1) else (return 0))
--> [return 1 , return 0]
--> [1 , 0]
```

オペレーションの動作は、ハンドラの定義によって自由に変えることができる。この特徴によって、計算を再利用しやすくなるという利点がある。例えば、以下のプログラムでハンドラは *coin* オペレーションの動作を、「継続に *true* を渡して再開する」と定義している。よってプログラム全体の結果は先ほどの例と違い、1となる。

```
handle {
  if (do coin()) then (return 1) else (return 0)
}
with {
  return x → x
  coin () k → k true
}
```

4.2 構文

この章で扱う、STLCを代数的で拡張した言語を本稿では λ_e と呼ぶことにする。この言語の定義は、Hillerström ら [9] の言語に基づいている。 λ_e の構文を図 4.1 に示す。 λ_e の項は、エフェクトを起こさない純粋な「値」と、エフェクトを起こす可能性のある「計算」の2つからなる。

まず型について説明する。値の型は、*Unit* 型と、*A* 型の値を受け取って *C* 型の計算を行う関数型 $A \rightarrow C$ からなる。エフェクトは、オペレーションの集合である。各オペレーションは引数の型と、結果の型の2つの情報を持つ。これをオペレーションのシグネチャという。計算の型 $A!E$ は、計算結果の値の型 *A* と、その計算が起こす可能性のあるエフェクト *E* のペアである。 $C \Longrightarrow D$ 型のハンドラは、*C* 型の計算を処理して *D* 型の計算に変換する。

次に値と計算とハンドラについて説明する。値は *unit* と変数と関数からなる。計算は関数適用、*return* 式、オペレーション呼び出し、*let* 束縛、エフェクトハンドリングからなる。*return* 式 *return V* は、値 *V* を

(値の型)	$A, B ::=$	$Unit \mid A \rightarrow C$
(エフェクト)	$E ::=$	$\{l : A \rightarrow B\}_{l \in \mathcal{L}}$
(計算の型)	$C, D ::=$	$A!E$
(ハンドラの型)	$F ::=$	$C \Rightarrow D$
(値)	$V, W ::=$	$unit \mid x \mid \lambda x. C$
(計算)	$M, N ::=$	$V W \mid return V \mid do l V \mid$ $let x = M in N \mid handle M with H$
(ハンドラ)	$H ::=$	$\{return x \rightarrow M\} \mid \{l p k \rightarrow M\} \uplus H$

Figure 4.1: λ_e の構文

結果として返す計算である。オペレーション呼び出し $do l V$ は、 l とラベル付けされたオペレーションを引数 V で呼び出す計算である。let 束縛 ($let x = M in N$) は、まず計算 M を実行し、結果の値を x に束縛して、 N を実行する計算である。エフェクトハンドリング ($handle M with H$) は、計算 M が起こす副作用をハンドラ H で処理する。ハンドラ H は、 $return$ の処理と、0 個以上のオペレーションの処理の定義の集合からなる。

4.3 型規則

値、計算、ハンドラの型判断はそれぞれ $\Gamma \vdash V : A$ 、 $\Gamma \vdash M : C$ 、 $\Gamma \vdash H : C \Rightarrow D$ の形をとる。値の型判断は、「型環境 Γ のもとで値 V は A 型を持つ」ことを意味する。計算とハンドラの型判断についても同様である。

型規則を図 4.2 に示す。値や関数適用の型付けは STLC と同様である。規則 T-RETURN は任意のエフェクト E について、返す値の型を結果の型とする規則である。規則 T-DO は、エフェクト E に含まれるオペレーション $l : A \rightarrow B$ を A 型の値で呼び出すと、結果として B 型の値が返ることを表す。規則 T-LET は、 $A!E$ 型の計算 M の結果、すなわち A 型の値を x に束縛したときの N の型を検査する。規則 T-HANDLE は、 C 型の計算が $C \rightarrow D$ 型のハンドラによって D 型の計算になることを表す。ハンドラの型規則 T-HANDLER は、 $return$ と計算の型 C 中のオペレーションとシグネチャ $\{l p k \rightarrow N_l\}_l$ を D 型の計算で処理するように定義されたハンドラに $C \Rightarrow D$ 型を付ける。

値の型規則

$$\Gamma \vdash \mathit{unit} : \mathit{Unit} \text{ (T-UNIT)} \quad \frac{x : A \in \Gamma}{\Gamma \vdash x : A} \text{ (T-VAR)} \quad \frac{\Gamma, x : A \vdash M : C}{\Gamma \vdash \lambda x.M : A \rightarrow C} \text{ (T-ABS)}$$

計算の型規則

$$\frac{\Gamma \vdash V : A \rightarrow C \quad \Gamma \vdash W : A}{\Gamma \vdash V W : C} \text{ (T-APP)} \quad \frac{\Gamma \vdash V : A}{\Gamma \vdash \mathit{return} V : A!E} \text{ (T-RETURN)}$$

$$\frac{(l : A \rightarrow B) \in E \quad \Gamma \vdash V : A}{\Gamma \vdash \mathit{do} l V : B!E} \text{ (T-DO)} \quad \frac{\Gamma \vdash M : A!E \quad \Gamma, x : A \vdash N : B!E}{\Gamma \vdash \mathit{let} x = M \mathit{in} N : B!E} \text{ (T-LET)}$$

$$\frac{\Gamma \vdash M : C \quad \Gamma \vdash H : C \Rightarrow D}{\Gamma \vdash \mathit{handle} M \mathit{with} H : D} \text{ (T-HANDLE)}$$

ハンドラの型規則

$$\frac{H = \{\mathit{return} x \rightarrow M\} \uplus \{l p k \rightarrow N_l\}_l \quad C = A!l : A_l \rightarrow B!l \quad \Gamma, x : A \vdash M : D \quad \forall l. \Gamma, p : A_l, k : B_l \rightarrow D \vdash N_l : D}{\Gamma \vdash H C \Rightarrow D} \text{ (T-HANDLER)}$$

Figure 4.2: λ_e の型規則

4.4 CPS 意味論

代数的エフェクトとハンドラのための CPS 変換 [9] に基づいて、図 4.3 のように CPS 意味論を定義する。計算を評価する ξ_c は新たにハンドラ h を引数に受け取るようになる。ハンドラ h は、オペレーションのラベル、パラメーター、継続を受け取ってオペレーションを処理する。純粋な値はメタ言語の値に評価される。 $\mathit{return} V$ の評価は、継続に V の評価結果を渡して再開させる。 $\mathit{do} l V$ の評価は、ハンドラ h にラベル l 、引数 V 、継続 $\lambda x.\kappa x h$ を渡してオペレーションを処理する。 $\mathit{let} x = M \mathit{in} N$ の評価は、 M を評価して、その結果を環境 ρ に追加して N を評価する。この評価順序は ξ_c に渡す継続によって制御している。 $\mathit{handle} M \mathit{with} H$ の評価は、 H に基づいて生成された継続とハンドラで M を評価する。このときの継続とハンドラは、ハンドラ H の return 節の H_{ret} 、 $\mathit{operation}$ 節の

$$\begin{aligned}
\xi_v[\mathit{unit}]\rho &= \mathit{unit} \\
\xi_v[x]\rho &= \rho(x) \\
\xi_v[\lambda x.M] &= \lambda v.\lambda\kappa.\lambda h.\xi_c[M]\rho[v/x]\kappa h \\
\xi_c[V\ W]\rho &= (\xi_v[V]\rho)\ (\xi_v[W]\rho) \\
\xi_c[\mathit{return}\ V]\rho\kappa &= \kappa\ (\xi_v[V]\rho) \\
\xi_c[\mathit{do}\ l\ V]\rho\kappa h &= h(l, \xi_v[V]\rho, \lambda x.\kappa\ x\ h) \\
\xi_c[\mathit{let}\ x = M\ \mathit{in}\ N]\rho\kappa &= \xi_c[M]\rho(\lambda v.\xi_c[N]\rho[v/x]\kappa) \\
\xi_c[\mathit{handle}\ M\ \mathit{with}\ H]\rho\kappa h &= \xi_c[M]\rho(\xi_h[H_{ret}]) (\xi_h[H_{ops}]) \\
&\text{where } H_{ret} = \{\mathit{return}\ x \rightarrow N\} \\
&\quad H_{ops} = \{l\ p_l\ k_l \rightarrow N_l\}_l \\
&\quad \xi_h[H_{ret}] = \lambda x.\lambda h.\xi_c[N] \\
&\quad \xi_h[H_{ops}] = \lambda z.\mathit{case}\ z\ \mathit{with}\ \{(l, p, k) \rightarrow \xi_c[N_l]\rho[p/p_l][k/k_l]\} \\
\top[M] &= \xi_c[M]\emptyset(\lambda x.\lambda h.x)(\lambda z.())
\end{aligned}$$

Figure 4.3: λ_e の CPS 意味論

H_{ops} から作られる。

型システムと意味論を定義したので、 λ_e の型安全性を定義する。 λ_e の型安全性は、 $\Gamma \vdash M : A \neq \emptyset$ なる M に対して、 $\top[M]$ が A 型の値になることである。

4.5 構文の実装

ここでは、 λ_e を Agda で実装する方法を示す。

まず、型の定義にしたがって型の構文を以下のように定義する。**VTy**、**CTy**、**HTy** はそれぞれ値の型、計算の型、ハンドラの型を表す。**Sig** はエフェクトシグネチャを表し、エフェクトを表す **Eff** は、**Sig** のリストとして定義される。

```

-- Value types
data VTy : Set where
  Unit : VTy
  _=>_ : VTy -> CTy -> VTy

```

```

-- Effect signatures
data Sig : Set where
  op : VTy → VTy → Sig

-- Effects
Eff : Set
Eff = List Sig

-- Computation types
CTy : Set
CTy = VTy × Eff

-- Handler types
data HTy : Set where
  _⇒_ : CTy → CTy → HTy

```

次に、定義にしたがって値、計算、ハンドラの構文を以下のように定義する。STLCと同様に変数束縛にはPHOASを使用している。Doコンストラクタに渡すオペレーションラベルは、シグネチャのリストであるEff中のインデックスを表す型 \in で表現している。ハンドラのoperation節は、型レベルのmap関数であるAllを使って、エフェクト E 中の各シグネチャに対応する動作定義を受け取る。

```

mutual
  -- Values
  data Val (Var : VTy → Set) : VTy → Set where
    unit : Val Var Unit
    var  : ∀{T} → Var T → Val Var T
    fun  : ∀{A C} → (Var A → Cmp Var C) → Val Var (A ⇒ C)

  -- Computations
  data Cmp (Var : VTy → Set) : CTy → Set where
    app    : ∀{A C} → Val Var (A ⇒ C) → Val Var A → Cmp Var C
    Return : ∀{E A} → Val Var A → Cmp Var (A, E)
    Do     : ∀{E A B} → (op A B) ∈ E → Val Var A → Cmp Var (B, E)
    Let.In_ : ∀{E A B} → Cmp Var (A, E) →
      (Var A → Cmp Var (B, E)) → Cmp Var (B, E)
    Handle.With_ : ∀{C D} → Cmp Var C → Handler Var (C ⇒ D) → Cmp Var D

  -- Handlers

```

```

data Handler (Var : VTy → Set) : HTy → Set where
  return_/ops_ : ∀{A E C} →
    -- return clause
    (Var A → Cmp Var C) →
    -- operation clauses
    All (λ {(op A' B') →
      Var A' → (Var (B' ⇒ C)) → Cmp Var C }
    ) E →
    Handler Var ((A , E) ⇒ C)

```

4.6 型の変換

STLCの実装と同様に、型の変換を実装する。ここでは、値の型の変換 $\sim t$ と、計算の型の変換 $\sim c$ を定義する。 $\sim t$ は自明な型の変換である。 $\sim c$ による計算の型の変換先は、それが持つエフェクトに基づくCmp型の継続と、Hand型のハンドラを受け取り、答えの型 α の値を返す関数型である。 $\sim c$ の本来の型は $Cty \rightarrow Set_1$ であるが、Agdaの`{-# -type-in-type #-}`オプションを使うことで Set と Set_1 の違いを無視している。また、ハンドラの型Handの定義はAgdaの停止性検査を通過しないので、`{-# TERMINATING #-}`オプションを使って停止性検査をスキップしている。

```

mutual
  ~t : VTy → Set
  ~t Unit = ⊥
  ~t (A ⇒ C) = ~t A → ~c C

  ~c : CTy → Set
  ~c (A , E) = ∀{ α : Set } → Cont (A , E) α → Hand E α → α

  Cont : CTy → Set → Set
  Cont (A , E) α = ~t A → Hand E α → α

  {-# TERMINATING #-}
  Hand : Eff → Set → Set
  Hand E α = ∀{A B} → (op A B) ∈ E → ~t A → (~t B → α) → α

```

4.7 インタプリタ

ここでは3つのインタプリタを以下に定義する。`eval` はトップレベルの λ_e プログラムの評価関数であり、これが型安全インタプリタそのものである。`eval` の型は λ_e の型安全性の言明を表しており、その第一引数は空のエフェクトを持つ計算の型である。これは、トップレベルのプログラムがハンドラによって全ての副作用を処理する必要があることを示している。

`eval` は `evalv` と `evalc` の2つのインタプリタを使って定義される。`evalv` は値のための、`evalc` は純粋でないものを含む任意の計算のためのインタプリタである。

$$\begin{aligned} \text{eval} &: \forall\{A\} \rightarrow \text{Cmp } \sim t (A, []) \rightarrow \sim t A \\ \text{evalv} &: \forall\{A\} \rightarrow \text{Val } \sim t A \rightarrow \sim t A \\ \text{evalc} &: \forall\{C\} \rightarrow \text{Cmp } \sim t C \rightarrow \sim c C \end{aligned}$$

3つのインタプリタの本体の実装を以下に示す。STLCと同様 `evalv`、`evalc`、`eval` はそれぞれ 4.4 節で示した ξ_v 、 ξ_c 、 Υ の定義に対応する。

$$\begin{aligned} \text{evalv } \text{unit} &= \text{tt} \\ \text{evalv } (\text{var } x) &= x \\ \text{evalv } (\text{fun } f) &= \lambda x \rightarrow \text{evalc } (f x) \\ \\ \text{evalc } (\text{app } v1 v2) k &= \\ &\quad \text{let } f = \text{evalv } v1 \text{ in} \\ &\quad \text{let } v = \text{evalv } v2 \text{ in} \\ &\quad f v k \\ \text{evalc } (\text{Return } v) k &= \\ &\quad k \$ \text{evalv } v \\ \text{evalc } (\text{Do } \text{label } v) k h &= \\ &\quad h \text{label } (\text{evalv } v) (\lambda x \rightarrow k x h) \\ \text{evalc } (\text{Let } e \text{ In } f) k &= \\ &\quad \text{evalc } e \$ \lambda x \rightarrow \\ &\quad \text{evalc } (f x) k \\ \text{evalc } \{C = C\} (\text{Handle } e \text{ With } (\text{return } \text{ret } / \text{ops } \text{calls})) &= \\ &\quad \text{evalc } e \{ \alpha = \sim c C \} (\lambda x _ \rightarrow \text{evalc } (\text{ret } x)) (\lambda l p k \rightarrow \text{evalc } (\text{lookup } \text{calls } l p k)) \\ \\ \text{eval } c &= \text{evalc } c \text{ init-k pure-h} \\ \text{where} & \\ \text{init-k} &: \forall\{A\} \rightarrow \text{Cont } (A, []) (\sim t A) \\ \text{init-k } x _ &= x \end{aligned}$$

```
pure-h : ∀{A} → Hand [] (~t A)
pure-h ()
```

λ_e では、対象言語の CPS 意味論にハンドラが追加されたが、CPS 意味論、PHOAS の型付き AST、型の変換を適切に与えれば、今までと同様に CPS 意味論に直接対応する型安全インタプリタを記述できる。型の変換の実装の際に `{-# -type-in-type #-}` と `{-# TERMINATING #-}` の Agda オプションを使用したがる、これらを取り除くことは今後の課題である。

第5章 関連研究

5.1 型付き構文木と型安全インタプリタ

依存型 依存型言語を用いて Intrinsically-typed language を定義するという方法は以前から提唱されてきた。例えば、Altenkirch と Reus[1] は STLC のための型付き抽象構文木に対する型安全インタプリタを依存型言語で定義することを提唱した。

Tagless Final Tagless Final[5] とは、メタ言語上に対象言語を設計する際に用いられる統合的手法である。Tagless Final の利点は、型安全インタプリタと同様に対象言語とその評価を型安全に定義でき、対象言語のプログラムの型付けをメタ言語の型システムに委譲できる点である。さらに、Tagless Final は依存型などの発展的な型システムを必要とせず、Scala や Haskell などの通常のプログラミング言語上で実装できる。Tagless Final では HOAS を使用することができ、対象言語の変数束縛をメタ言語で表現できる。また、対象言語のプログラムの評価には、インタプリタ、CPS 変換、部分評価器などの実装をサポートしている。

5.2 高度な機能の型安全な実装

これまでに、高度な言語機能を持つ言語のための型安全インタプリタを簡潔に定義する方法とそのためのライブラリが提唱されてきた。

Poulsen ら [3] は、可変状態や動的束縛の機能を持つ言語を実装するためのライブラリを開発した。このライブラリは型安全な可変状態ストアに対する操作を行う命令を含んでおり、これによってインタプリタの簡潔な実装が可能になる。本研究と違い、対象言語の構文は de Bruijn インデックスを用いて実装されており、インタプリタにはモナドが使用されている。de Bruijn インデックスとは、変数束縛の表現のひとつであり、変数はその名前の代わりに、その変数を束縛する λ が何階層外側にあるかを表す数値を使う。例えば $\lambda x. \lambda y. x y$ は、de Bruijn インデックスを使うと $\lambda. \lambda. 1 0$ と表現される。PHOAS による実装との違いは、変数コンストラクタは参照先の値を受け取らず、型環境 Γ の要素へのポインタで束縛関係を表現しており、代入規則を定義する必要があるという点が挙げられる。

Poulsen らのインタプリタは、型環境と対応する、値環境を評価時に使用し、変数コンストラクタの評価は、引数のポインタで値環境中の値を取り出す。また、関数適用の評価においては、値環境に引数部分の評価結果を挿入してから、関数本体を評価することで変数への代入を表現している。

また、Rouvoet ら [14] は、線形型やセッション型を持つ言語を実装するためのライブラリを開発した。ライブラリは分離論理に基づいて型付き抽象構文木とインタプリタの実装を抽象化した。この抽象化によって実装から線形性の証明に関わる項を隠し、簡潔な実装を支援している。こちらもインタプリタにはモナドを使用しており、線形的な値の取り扱いを強制している。

第6章 まとめと今後の課題

本論文では、限定継続命令を持つ言語 $\lambda_{s/r}$ と、代数的エフェクトとハンドラを持つ言語 λ_e の型安全インタプリタを定義した。これらのインタプリタは、Agda の型検査を通し、それぞれの対象言語の型安全性を証明している。

$\lambda_{s/r}$ と λ_e の型安全インタプリタは、特別な抽象化を必要とせずとも簡潔な実装は可能であった。これは、型付き抽象構文木を PHOAS で実装することによって、CPS 意味論の率直な書き下しとして、インタプリタを実装することができたからである。この方法は try-catch などの継続を操作するようなエフェクトに対しても有効であると考えられる。

また、本論文のインタプリタは、継続やハンドラを明示的に引数に受け取っていた。これらを隠蔽し、継続やハンドラに関する操作を抽象化するモナドを開発できると考えられる。このモナドにより継続やハンドラを直接扱うことなく、手続き的に意味論を記述できることが期待できる。

本論文では、 λ_e の実装の際に、Agda のオプションを用いて、インタプリタや型の変換は停止するという仮定 (`{-# TERMINATING #-}`) と、型のレベルの違いを無視できるという仮定 (`{-# -type-in-type #-}`) を置いた。これらの仮定を解消することは今後の課題である。

停止しない関数は論理における矛盾に対応するため、Agda の型検査は関数が停止することを要求する。`{-# TERMINATING #-}` を付けた関数に対しては、Agda の停止性検査を省略でき、停止しないことによる矛盾を無視できる。

現在 Agda は *Set* 型にレベルをつけている。Russell のパラドックスは、すべての集合の集合は正しい集合ではないということを意味する。もしすべての集合の集合 U が存在すれば、その部分集合 $A = \{S \mid S \notin S\}$ について $A \notin A \Leftrightarrow A \in A$ という矛盾が導ける。同様に、Girard のパラドックスは、Agda の型理論において $Set : Set$ は矛盾しているということを示した。例えば、 $Set : Set_1$ であるし、 $Set_1 : Set_2$ である。レベルに上限はない。`{-# -type-in-type #-}` は、これらのレベルを無視し、 $Set : Set$ を認めるオプションである。このようなオプションは取り除くことが好ましい。

今後は、エフェクトを持つ言語の型安全インタプリタをモナドで記述し

ライブラリ化する。また、コエフェクトを持つ言語のための型安全インタプリタ実装を分析するために、quantitative types[2]などの機能を持つ言語のためのインタプリタを実装する。

6.1 トレースエフェクト

イベントトレースとは、プログラムの実行中に起こる副作用などのイベントの順序である。イベントトレースを管理することにより、安全なロック操作などの、イベントの順序が重要なプログラムの性質を静的に強制することができる。トレースエフェクト [15] は、エフェクト型システムとイベントトレースの検査を組み合わせたシステムであり、コンパイル時に自動でプログラムのイベントトレースを予測し、その性質を静的に検証できる。トレースエフェクトの言語モデルはラムダ計算に基づいている。この言語の型安全インタプリタを実装するために、型付き抽象構文木にイベントトレースに関する証明項を追加する。

6.2 コエフェクト計算

コエフェクト計算は、コエフェクトと呼ばれる様々なリソースの使用状況を分析するための計算体系である。例えば、線形型システム [10] はコエフェクト計算の一種である。線形型システムでは、変数の使用回数を追跡することにより、各変数とそのスコープ内で正確に一度だけ使用されることを保証する型システムである。これにより、変数が使用されない、もしくは二回以上使用されるプログラムを不正なプログラムとして検出することができる。

quantitative type [2] は、線形型より詳細に変数の使用回数を追跡できる型システムである。quantitative type は、線形型と依存型を組み合わせることで、プロトコルが要求する操作の順番、すなわちどの操作がいつ許可されるのかを型で表現する。これにより、メモリ安全性、プロトコルの正しさを推論できる。quantitative type theory では、各変数の束縛に $0, 1, \omega$ の多重度を割り当てることで変数の使用回数を管理する。多重度 0 の変数は型の中で自由に扱えるが、それ以外の場所で使うことができず、多重度 1 の変数は正確に一度だけ使用され、多重度 ω の変数は無制限に使うことができる。

quantitative type を持つ言語のための型安全インタプリタを実装するために、変数の多重性を再現するような証明項を追加する。

Bibliography

- [1] Altenkirch, T. and Reus, B.: Monadic Presentations of Lambda Terms Using Generalized Inductive Types, *CSL '99* (1999).
- [2] Atkey, R.: Syntax and Semantics of Quantitative Type Theory, *LICS '18* (2018).
- [3] Bach Poulsen, C., Rouvoet, A., Tolmach, A., Krebbers, R. and Visser, E.: Intrinsically-Typed Definitional Interpreters for Imperative Languages, *POPL* (2017).
- [4] Bauer, Andrej and Pretnar, M.: An Effect System for Algebraic Effects and Handlers (2013).
- [5] Carette, J., Kiselyov, O. and Shan, C.-c.: Finally Tagless, Partially Evaluated: Tagless Staged Interpreters for Simpler Typed Languages, *J. Funct. Program.* (2009).
- [6] Chlipala, A.: Parametric Higher-Order Abstract Syntax for Mechanized Semantics, *ICFP '08* (2008).
- [7] Danvy, O. and Filinski, A.: A Functional Abstraction of Typed Contexts (1989).
- [8] Danvy, O. and Filinski, A.: Abstracting Control, *LFP '90* (1990).
- [9] Hillerström, D., Lindley, S., Atkey, R. and Sivaramakrishnan, K. C.: Continuation Passing Style for Effect Handlers, *FSCD 2017* (2017).
- [10] Morris, J. G.: The Best of Both Worlds: Linear Functional Programming without Compromise, *ICFP* (2016).
- [11] Norell, U.: *Towards a practical programming language based on dependent type theory*, Vol. 32 (2007).
- [12] Plotkin, G. D.: Call-by-name, call-by-value and the λ -calculus, *Theoretical computer science* (1975).

- [13] Pretnar, M.: An Introduction to Algebraic Effects and Handlers. Invited Tutorial Paper, *Electron. Notes Theor. Comput. Sci.*
- [14] Rouvoet, A., Bach Poulsen, C., Krebbers, R. and Visser, E.: Intrinsically-Typed Definitional Interpreters for Linear, Session-Typed Languages, CPP 2020 (2020).
- [15] Skalka, C., Smith, S. and Van horn, D.: Types and Trace Effects of Higher Order Programs (2008).
- [16] Tsuyama, S., Cong, Y. and Masuhara, H.: Intrinsically-Typed Interpreters for Effectful and Coeffectful Languages, Presented at the first Workshop on the Implementation of Type Systems (WITS 2022) (2022).